Programare declarativă

Functori, functori aplicativi, monoizi[1]

Traian Florin Șerbănuță

Departamentul de Informatică, FMI, UNIBUC
traian.serbanuta@fmi.unibuc.ro

9 decembrie 2016

[1] bazat pe Learn You a Haskell for Great Good

Cutii și computații

Tipuri parametrizate — „cutii”

Idee

O clasă largă de tipuri parametrizate pot fi gândite ca „cutii”, recipiente care pot conține elemente de tipul dat ca argument.

Exemple

- Clasa de tipuri opțiune asociază unui tip a, tipul **Maybe** a
  - cutii goale: **Nothing**
  - cutii care țin un element x de tip a: **Just** x
- Clasa de tipuri listă asociază unui tip a, tipul [a]
  - cutii care țin 0, 1, sau mai multe elemente de tip a: [1, 2, 3], [], [5]

Tipuri parametrizate — „cutii”

Idee

O clasă largă de tipuri parametrizate pot fi gândite ca „cutii”, recipiente care pot conține elemente de tipul dat ca argument.

Exemplu: tip de date pentru arbori binari

```
data Arbore a = Nil
              | Nod a Arbore Arbore
```

- Un arbore este o „cutie” care poate ține 0, 1, sau mai multe elemente de tip a:
  Nod 3 Nil (Nod 4 (Nod 2 Nil Nil) Nil), Nil, Nod 3 Nil Nil

Generalizare: Tipuri parametrizate — „computații”

Idee

O clasă largă de tipuri parametrizate pot fi gândite ca „contexte computaționale”: computații care, atunci când se execută, pot produce rezultate de tipul dat ca argument.

Exemple

- **Maybe** a descrie rezultate de computații deterministe care pot eșua
  - computații care eșuează: **Nothing**
  - computații care produc un element de tipul dat: **Just** 4
- [**Int**] descrie liste de rezultate posibile ale unor computații nedeterministe
  - care pot produce oricare dintre rezultatele date: [1, 2, 3], [], [5]

Tipuri parametrizate — „computații”

Idee

O clasă largă de tipuri parametrizate pot fi gândite ca „contexte computaționale”: computații care, atunci când se execută, pot produce rezultate de tipul dat ca argument.

Exemple

- **IO** a descrie computații care atunci când se execută produc rezultate de tip a
  - **getLine** :: **IO String**, **getChar** :: **IO Char**
- **Either** e a descrie rezultate de tip a ale unor computații deterministe care pot eșua cu o eroare de tip *e*
  - **Right** 5 :: **Either** e **Int** reprezintă rezultatul unei computații reușite
  - **Left** "OOM":: **Either String** a reprezintă o excepție de tip **String**

Tipuri parametrizate — „computații”

Idee

O clasă largă de tipuri parametrizate pot fi gândite ca „contexte computaționale”: computații care, atunci când se execută, pot produce rezultate de tipul dat ca argument.

Exemplu: tipul funcțiilor de sursă dată

- t –> a descrie computații care atunci când primesc o intrare de tip t produc un rezultat de tip a
  - (++ "!") :: **String** –> **String** este o computație care dat fiind un șir, îi adaugă un semn de exclamare
  - **length** :: **String** –> **Int** este o computație care dat fiind un șir, îi prduce lungimea acestuia
  - **id** :: **String** –> **String** este o computație care produce șirul dat ca argument

Clase de tipuri pentru cutii și computații?

Întrebare

Care sunt trăsăturile comune ale acestor tipuri parametrizate care pot fi gândite intuitiv ca cutii care conțin elemente / computații care produc rezultate?

Problemă

Putem proiecta clase de tipuri care descriu funcționalități comune tuturor acestor tipuri?

Functori

Problemă

Formulare cu cutii

Dată fiind o funcție `f :: a -> b` și o cutie ca care conține elemente de tip a, vreau să să obțin o cutie cb care conține elemente de tip b obținute prin transformarea elementele din cutia ca folosind funcția `f` (și doar atât!)

Formulare cu computații

Dată fiind o funcție `f :: a -> b` și o computație ca care produce rezultate de tip a, vreau să să obțin o computație cb care produce rezultate de tip b obținute prin transformarea rezultatelor produse de computația ca folosind funcția `f` (și doar atât!)

Exemplu — liste

Dată fiind o funcție `f :: a -> b` și o listă *la* de elemente de tip a, vreau să să obțin o lista de elemente de tip b transformând fiecare element din *la* folosind funcția `f` (și doar atât!)

Clasa de tipuri Functor

Definiție

```
class Functor f where
  fmap :: (a -> b) -> f a -> f b
```

Dată fiind o funcție f :: a -> b și ca :: f a, fmap produce cb :: f b obținută prin transformarea rezultatelor produse de computația ca folosind funcția f (și doar atât!)

Instanță pentru liste

```
instance Functor [] where
  fmap = map
```

Clasa de tipuri Functor

Instanțe

```
class Functor f where
  fmap :: (a -> b) -> f a -> f b
```

Instanță pentru tipul optiune fmap :: (a –> b) –> **Maybe** a –> **Maybe** b

Instanță pentru tipul arbore fmap :: (a –> b) –> Arbore a –> Arbore b

Clasa de tipuri Functor

Instanțe

```
class Functor f where
  fmap :: (a -> b) -> f a -> f b
```

Instanță pentru tipul optiune fmap :: (a -> b) -> **Maybe** a -> **Maybe** b

```
instance Functor Maybe where
  fmap f Nothing = Nothing
  fmap f (Just x) = Just (f x)
```

Instanță pentru tipul arbore fmap :: (a -> b) -> Arbore a -> Arbore b

```
instance Functor Arbore where
  fmap f Nil = Nil
  fmap f (Nod x l r) = Nod (f x) (fmap f l) (fmap f r)
```

Clasa de tipuri Functor

Instanțe

```
class Functor f where
  fmap :: (a -> b) -> f a -> f b
```

Instanță pentru tipul eroare fmap :: (a –> b) –> **Either** e a –> **Either** e b

Instanță pentru tipul funcție fmap :: (a –> b) –> (t –> a) –> (t –> b)

Clasa de tipuri Functor

Instanțe

```
class Functor f where
  fmap :: (a -> b) -> f a -> f b
```

Instanță pentru tipul eroare fmap :: (a -> b) -> **Either** e a -> **Either** e b

```
instance Functor (Either e) where
    fmap _ (Left x) = Left x
    fmap f (Right y) = Right (f y)
```

Instanță pentru tipul funcție fmap :: (a -> b) -> (t -> a) -> (t -> b)

```
instance Functor (->) a where
  fmap f g = f . g  -- sau, mai simplu, fmap = (.)
```

Clasa de tipuri Functor

Instanțe

```
class Functor f where
  fmap :: (a -> b) -> f a -> f b
```

Instanță pentru tipul I/O fmap :: (a –> b) –> **IO** a –> **IO** b

- Folosind notația **do**

- Folosind operatorul de legare

Clasa de tipuri Functor

Instanțe

```
class Functor f where
  fmap :: (a -> b) -> f a -> f b
```

Instanță pentru tipul I/O fmap :: (a –> b) –> **IO** a –> **IO** b

- Folosind notația **do**

```
instance Functor IO where
  fmap f ioa = do
    x <- ioa
    return (f x)
```

- Folosind operatorul de legare

```
fmap f ioa = ioa >>= (\ x -> return (f x))
```

- Sau, mai scurt,

```
fmap f ioa = ioa >>= (return . f)
```

Exemple

```
Main> fmap (*2) [1..3]

Main> fmap (*2) (Just 200)

Main> fmap (*2) Nothing

Main> fmap (*2) (+100) 4

Main> fmap (*2) (Right 6)

Main> fmap (*2) (Left 1)

Main> fmap (show . (*2) . read) getLine >>= putStrLn
```

Exemple

```
Main> fmap (*2) [1..3]
[2,4,6]
Main> fmap (*2) (Just 200)
Just 400
Main> fmap (*2) Nothing
Nothing
Main> fmap (*2) (+100) 4
208
Main> fmap (*2) (Right 6)
Right 12
Main> fmap (*2) (Left 135)
Left 135
Main> fmap (show . (*2) . read) getLine >>= putStrLn
123
246
```

Proprietăți ale functorilor

- Argumentul f al lui **Functor** f definește o transformare de tipuri
  - f a este tipul a transformat prin functorul f
- fmap definește transformarea corespunzătoare a funcțiilor
  - fmap :: (a –> b) –>(f a –> f b)

Contractul lui fmap

- fmap f ca e obținută prin transformarea rezultatelor produse de computația ca folosind funcția f (și doar atât!)
- Abstractizat prin două legi:

identitate fmap **id** == **id**

compunere fmap (f . g) == fmap f . fmap g

Categorii și Functori

Categorii

O categorie $\mathbb{C}$ este dată de:

- O clasă $|\mathbb{C}|$ a obiectelor
- Pentru oricare două obiecte $A, B \in |\mathbb{C}|$,
  o mulțime $\mathbb{C}(A, B)$ a săgeților „de la A la B”
  $f \in \mathbb{C}(A, B)$ poate fi scris ca $f : A \to B$
- Pentru orice obiect A o săgeată $id_A : A \to A$ numită identitatea lui A
- Pentru orice obiecte A, B, C, o operație de compunere a săgeților
  $\circ : \mathbb{C}(B, C) \times \mathbb{C}(A, B) \to \mathbb{C}(A, C)$



Bartosz Milewski — Category: The Essence of Composition

- Compunerea este asociativă și are element neutru id

Exemplu: Categoria $\mathbb{S}$et

- Obiecte: mulțimi
- Săgeți: funcții
- Identități: Funcțiile identitate
- Compunere: Compunerea funcțiilor

Exemplu: Categoria $\mathbb{H}$ask

- Obiectele: tipuri
- Săgețiile: funcții între tipuri

```
f :: A -> B
```

- Identități: funcția polimorfică **id**

```
Prelude> :t id
id :: a -> a
```

- Compunere: funcția polimorfică (.)

```
Prelude> :t (.)
(.) :: (b -> c) -> (a -> b) -> a -> c
```

Subcategorii ale lui Hask date de tipuri parametrizate

- Obiecte: o clasă restânsă de tipuri din $|\mathbb{H}\mathrm{ask}|$
  - Exemplu: tipuri de forma [a]
- Săgeți: toate funcțiile din $\mathbb{H}\mathrm{ask}$ între tipurile obiecte
  - Exemple: **concat** :: [[a]] –> [a], **words** :: [**Char**] –> [**String**], **reverse** :: [a] –> [a]

Exemple

Liste	obiecte: tipuri de forma [a]
Optiuni	obiecte: tipuri de forma Maybe a
Arbori	obiecte: tipuri de forma Arbore a
Comenzi I/O	obiecte: tipuri de forma IO a
Funcții de sursă t	obiecte: tipuri de forma t -> a

De ce categorii?

(Des)compunerea este esența programării

- Am de rezolvat problema P
- O descompun în subproblemele $P_1, \ldots P_n$
- Rezolv problemele $P_1, \ldots P_n$ cu programele $p_1, \ldots p_n$
  - Eventual aplicând recursiv procedura de față
- Compun rezolvările $p_1, \ldots p_n$ într-o rezolvare p pentru problema inițială

Categoriile rezolvă problema compunerii

- Ne forțează să abstractizăm datele
- Se poate acționa asupra datelor doar prin săgeți (metode?)
- Forțează un stil de compunere independent de structura obiectelor

Functori

Date fiind două categorii $\mathbb{C}$ și $\mathbb{D}$, un functor $F : \mathbb{C} \to \mathbb{D}$ este dat de

- O funcție $F : |\mathbb{C}| \to |\mathbb{D}|$ de la obiectele lui $\mathbb{C}$ la cele ale lui $\mathbb{D}$
- Pentru orice $A, B \in |\mathbb{C}|$, o funcție $F : \mathbb{C}(A, B) \to \mathbb{D}(F(A), F(B))$
- Compatibilă cu identitățile și cu compunerea
  - $F(id_A) = id_{F(A)}$ pentru orice A
  - $F(g \circ f) = F(g) \circ F(f)$ pentru orice $f : A \to B, g : B \to C, h = g \circ f$





Bartosz Milewski — Functors

Functori în Haskell

În general un functor $F : \mathbb{C} \to \mathbb{D}$ este dat de

- O funcție $F : |\mathbb{C}| \to |\mathbb{D}|$ de la obiectele lui $\mathbb{C}$ la cele ale lui $\mathbb{D}$
- Pentru orice $A, B \in |\mathbb{C}|$, o funcție $F : \mathbb{C}(A, B) \to \mathbb{D}(F(A), F(B))$
- Compatibilă cu identitățile și cu compunerea
  - $F(id_A) = id_{F(A)}$ pentru orice A
  - $F(g \circ f) = F(g) \circ F(f)$ pentru orice $f : A \to B, g : B \to C, h = g \circ f$

În Haskell o instanță **Functor** f este dată de

- Un tip f a pentru orice tip a (deci f trebuie sa fie tip parametrizat)
- Pentru orice două tipuri a și b, o funcție

```
fmap :: (a -> b) -> (f a -> f b)
```

- Compatibilă cu identitățile și cu compunerea

```
fmap id == id
fmap (g . f) == fmap g . fmap f
```

pentru orice f :: a -> b și g :: b -> c

Functori aplicativi

Problemă

- Folosind fmap putem transforma o funcție h :: a –> b într-o funcție între cutii/computații fmap h :: f a –> f b
- Dar ce se întâmplă dacă avem o funcție cu mai multe argumente E.g., cum trecem de la h :: a –> b –> c la h' :: f a –> f b –> f c
- putem încerca să folosim fmap

Problemă

- Folosind fmap putem transforma o funcție h :: a –> b într-o funcție între cutii/computații fmap h :: f a –> f b
- Dar ce se întâmplă dacă avem o funcție cu mai multe argumente E.g., cum trecem de la h :: a –> b –> c la h' :: f a –> f b –> f c
- putem încerca să folosim fmap
- Dar, deoarece h :: a –> (b –> c), avem că fmap h :: f a –> f (b –> c)
- Putem aplica fmap h la o valoare fa :: f a și obținem fmap h fa :: f (b –> c)
- Problemă: Cum transformăm o cutie care conține o funcție într-o funcție între cutii
- Dacă avem asta, putem transforma funcții cu oricâte argumente.

Clasa de tipuri Applicative

Definiție

```
class Functor f => Applicative f where
  pure :: a -> f a
(<*>) :: f (a -> b) -> f a -> f b
```

- Orice instanță a lui Applicative trebuie să fie instanță a lui **Functor**
- pure transformă o valoare într-o computație minimală care are acea valoare ca rezultat, și nimic mai mult!
- (<*>) ia o computație care produce funcții și o computație care produce argumente pentru funcții și obține o computație care produce rezultatele aplicării funcțiilor asupra argumentelor

Proprietate importantă

- fmap f x == pure f <*> x
- Se definește operatorul (<$>) prin (<$>) = fmap

Clasa de tipuri Applicative

Instanțe

```
class Functor f => Applicative f where
    pure :: a -> f a
  (<*>) :: f (a -> b) -> f a -> f b
```

Instanță pentru tipul opțiune

Instanță pentru tipul eroare

Clasa de tipuri Applicative

Instanțe

```
class Functor f => Applicative f where
    pure :: a -> f a
  (<*>) :: f (a -> b) -> f a -> f b
```

Instanță pentru tipul opțiune

```
instance Applicative Maybe where
  pure = Just
  Nothing <*> _ = Nothing
  Just f  <*> x = fmap f x
```

Instanță pentru tipul eroare

```
instance Applicative (Either a) where
  pure = Right
  Left e <*> _ = Left e
  Right f  <*> x = fmap f x
```

Clasa de tipuri Applicative

Instanțe

```
class Functor f => Applicative f where
    pure :: a -> f a
  (<*>) :: f (a -> b) -> f a -> f b
```

Instanță pentru tipul computațiilor nedeterministe (liste)

Instanță pentru tipul computațiilor I/O

Clasa de tipuri Applicative

Instanțe

```
class Functor f => Applicative f where
    pure :: a -> f a
  (<*>) :: f (a -> b) -> f a -> f b
```

Instanță pentru tipul computațiilor nedeterministe (liste)

```
instance Applicative [] where
  pure x = [x]
  fs  <*> xs = [f x | f <- fs, x <- xs]
```

Instanță pentru tipul computațiilor I/O

```
instance Applicative IO where
  pure = return
  iof <*> iox = do
    f <- iof
    x <- iox
    return (f x)
```

Clasa de tipuri Applicative

Instanțe

```
class Functor f => Applicative f where
    pure :: a -> f a
  (<*>) :: f (a -> b) -> f a -> f b
```

Instanță pentru tipul funcțiilor de sursă dată

Clasa de tipuri Applicative

Instanțe

```
class Functor f => Applicative f where
    pure :: a -> f a
  (<*>) :: f (a -> b) -> f a -> f b
```

Instanță pentru tipul funcțiilor de sursă dată

```
instance Applicative ((->) t) where
  pure :: a -> (t -> a)
  pure x = \ _ -> x
  (<*>) :: (t -> (a -> b)) -> (t -> a) -> (t -> b)
  f  <*> g = \ x -> f x (g x)
```

Clasa de tipuri Applicative

Exemple

```
Main> pure "Hey" :: [String]

Main> pure "Hey" :: Maybe String

Main> [(*0),(+100),(^2)] <*> [1,2,3]

Main> [(+),(*)] <*> [1,2] <*> [3,4]

Main> (++) <$> ["ha","heh","hm"] <*> ["?","!","."]

Main> filter (>50) $ (*) <$> [2,5,10] <*> [8,10,11]

Main> (++) <$> getLine <*> getLine >>= putStrLn
hello
 world!
```

ZipList

Idee: în loc să privim listele ca computații nedeterministe, le privim ca fluxuri de date.

```
newtype ZipList a = ZipList { getZipList :: [a]}

instance Functor ZipList where
  fmap f (ZipList xs) = ZipList (fmap f xs)

instance Applicative ZipList where
  pure x = repeat x
  ZipList fs <*> ZipList xs =
    ZipList (zipWith (\ f x -> f x) fs xs)
```

ZipList

Exemple

```
Main> getZipList $ (+) <$> ZipList [1,2,3] <*> ZipList
    [100,100,100]
[101,102,103]
Main> getZipList $ (+) <$> ZipList [1,2,3] <*> ZipList
    [100,100..]
[101,102,103]
Main> getZipList $ max <$> ZipList [1,2,3,4,5,3] <*> ZipList
     [5,3,1,2]
[5,3,3,4]
Main> getZipList $ (,,) <$> ZipList "dog" <*> ZipList "cat"
    <*> ZipList "rat"
[('d','c','r'),('o','a','a'),('g','t','t')]
```

Proprietăți ale functorilor aplicativi

identitate `pure id <*> v = v`

compoziție `pure (.) <*> u <*> v <*> w = u <*> (v <*> w)`

homomorfism `pure f <*> pure x = pure (f x)`

interschimbare `u <*> pure y = pure ($ y) <*> u`

Consecință: `fmap f x == f <$> x == pure f <*> x`

Monoizi

Monoizi

Data.Monoid.Monoid

```
class Monoid m where
    mempty  :: m
    mappend :: m -> m -> m
```

Monoidul listelor

```
instance Monoid [a] where
    mempty  = []
    mappend = (++)
```

Monoide booleene

Monoidul conjunctiv

Data.Monoid.All

```
newtype All = All { getAll :: Bool }

instance Monoid All where
        mempty = All True
        All x `mappend` All y = All (x && y)
```

Monoidul disjunctiv

Data.Monoid.Any

```
newtype Any = Any { getAny :: Bool }

instance Monoid Any where
        mempty = Any False
        Any x `mappend` Any y = Any (x || y)
```

Monoide numerice

Monoidul aditiv — Data.Monoid.Sum

```
newtype Sum a = Sum { getSum :: a }

instance Num a => Monoid (Sum a) where
        mempty = Sum 0
        Sum x `mappend` Sum y = Sum (x + y)
```

Monoidul multiplicativ — Data.Monoid.Product

```
newtype Product a = Product { getProduct :: a }

instance Num a => Monoid (Product a) where
        mempty = Product 1
        Product x `mappend` Product y = Product (x * y)
```

La ce sunt buni Monoizii?

Data.Foldable.foldMap

- Agregare într-un monoid:

```
foldMap :: Monoid m => (a -> m) -> [a] -> m
foldMap f = foldr (mappend . f) mempty
```

Exemple

```
concat :: [[a]] -> [a]
concat = foldMap id
```

```
all :: [Bool] -> Bool
all = getAll . foldMap All
```

```
any :: [Bool] -> Bool
any = getAny . foldMap Any
```

```
sum :: Num a => [a] -> a
sum = getSum . foldMap Sum
```

```
product :: Num a => [a] -> a
product =
  getProduct . foldMap Product
```

Monoidul endomorfismelor

```
newtype Endo a = Endo { appEndo :: a -> a }

instance Monoid (Endo a) where
        mempty = Endo id
        Endo f `mappend` Endo g = Endo (f . g)
```

- Mulțimea funcțiilor din A în A are structură de monoid

Relația între foldr și foldMap

foldMap în funcție de foldr

```
foldMap :: Monoid m => (a -> m) -> [a] -> m
foldMap f = foldr (mappend . f) mempty
```

foldr în funcție de foldMap

```
foldr :: (a -> b -> b) -> b -> [a] -> b
foldr f z t = appEndo (foldMap (Endo . f) t) z
```

Observație: f :: a -> (b -> b) translatează un element din a într-un endomorfism peste b